

ガロア群が G_{48} (またはその部分群) の方程式の厳密解 with *Mathematica* 14

例: $f(x) = x^6 + x^4 + 2x^3 + 3x^2 + 6x + 4$

2025 年5月31日 by mixedmoss

```
In[ ]:= ClearAll["Global`*"]
```

§1. ガロア群 G_{48} について

既約で可解な6次方程式のガロア群 Gal は大きく分けて、
3つのグループに分かれます。(添え数字は位数を表します.)

- G_{72} グループ: $G_{72}, \Gamma_{36}, \Gamma_{36}, G_{18}$ (G_{72} の部分群だが G_{48} の部分群でない)
- G_{48} グループ: $G_{48}, G_{24}, \Gamma_{24}, H_{24}, \Gamma_{12}$ (G_{48} の部分群 だが G_{72} の部分群でない)
- G_{12} グループ: G_{12}, C_6, H_6 (G_{72} と G_{48} の共通の部分群で G_{12} の部分群となる)

G_{48} の分解列は、例えば $G_{48} \triangleright \Gamma_{24} \triangleright \Gamma_{12} \triangleright K_4 \triangleright C_2 \triangleright \{\text{id}\}$ (K_4 はクラインの4元群)

G_{48} は定まった名前はないようですが、 $G_{48} \simeq C_2 \times S_4$ で、例えば

$$G_{48} = \langle \sigma, \tau, \mu \rangle, \quad \sigma = (1, 3, 5, 2, 4, 6), \quad \tau = (1, 2), \quad \mu = (1, 3, 2, 4)$$

です。ガロア群が G_{48} グループとなる方程式の例としては

「 $f(x) = (g(x))^3 + a(g(x))^2 + b g(x) + c$ (a, b, c は有理数, g は2次式)」
があります。例えば

「 $f(x) = (x^2 + 1)^3 + (x^2 + 1)^2 + (x^2 + 1) + 2 = x^6 + 4x^4 + 6x^2 + 5$ 」のガロア群は G_{48} です。

この例は、「3次方程式 $\rightarrow$ 2次方程式」の順で比較的簡単に解くことができます。

これと同様に「3次方程式 $\rightarrow$ 2次方程式」の順で、

$\text{Gal} = G_{48}$ やグループ G_{48}, G_{12} の方程式は解くことが可能です。

因みに $\text{Gal} = G_{48}$ の方程式は 可解で既約な5次方程式の約4割を占めており、

その内なんと8割ほどが上の様な形に簡単に直ります。

(但し f がmonicかつ整数係数でその絶対値が10以下の場合)

ここでは上の様な簡単な形にならない $\text{Gal} =$

G_{48} の方程式の例として「例: $f(x) = x^6 + x^4 + 2x^3 + 3x^2 + 6x + 4$ 」を取り上げます。

なお、ガロア群について詳しくは `sexticEquationsGaloisGroup.nb` をご覧ください。

§2. ガロア分解式 (Galois Resolvent)

5次方程式に関するDummit氏の論文(1991年)の, $f_{20}(x)=(x-\theta_1)(x-\theta_2)(x-\theta_3)(x-\theta_4)(x-\theta_5)(x-\theta_6)$ がその例です. ただこの論文では Galois Resolvent は $f_{20}(x)$ の一つだけで, また Galois Resolvent という言葉は使われていません. 6次方程式に関するHagedorn氏の論文(2000年)では, Galois Resolvent という言葉が与えられて, かつ10個ほど使われています. 一方「Minnesota Journal of Undergraduate Mathematics vol1 #1(2015年)」では absolute Resolvent という名前になっているので, 結局, 定まった名前はないようです. ガロア理論の入門書(数学書)には載っていないですが, 歴史は古く, Emmy Noether 氏が1915年の論文で触れたようです. Noether 氏がどのようなことを書いたのかは全く知りませんが, コンピューターを使って解を計算したりガロア群を求めるにはとても有効です. 私は今まで5次方程式や7次方程式を解くときは, 原始元 v による f の解の表現を使っていたのですが, $\text{Gal} = G_{72}$ の時は SageMath を使っても正しい式が見つからず, やむなく Galois Resolvent を使いました. その時, 非常に有用な事が分かったので G_{48} の場合も Galois Resolvent を使うことにしました.

まず「 $G_{48} = \langle \sigma, \tau, \mu \rangle$, $\sigma = (1, 3, 5, 2, 4, 6)$, $\tau = (1, 2)$, $\mu = (1, 3, 2, 4)$ 」によって不変な式を1つ見つけます. 例えば $\theta = x_1 x_2 + x_3 x_4 + x_5 x_6$ がそれに当たります. 次に $[S_6: G_{48}] = 6!/48 = 15$ なので, S_6 は G_{48} の15個の剰余類に分けられます. その代表元の集合を Σ_{15} とし, θ を Σ_{15} の全ての要素で置換して作った積 $p(x) = \prod (x - \theta_{\Sigma_{15}})$ は, $x_1 \sim x_6$ の対称式となり, $p(x)$ は有理数係数の多項式となります. そして $p(x)$ の15個の解のうちちょうど一個が θ となる訳ですが, θ は G_{48} によって不変なので, 有理数です. 即ち $p(x)$ は有理数解を少なくとも1つ持ちます. しかし可解な5次方程式の時は, 有理数解はただ1つだったのですが, 6次方程式では異なります. さてまずは Σ_{15} ($\{x_1, x_3, x_5, x_2, x_4, x_6\}$ の置換としての表現)を作ります.

```
Σ15 = {{x1, x2, x3, x4, x5, x6}, {x1, x2, x3, x5, x4, x6},
      {x1, x2, x3, x6, x4, x5}, {x1, x3, x2, x4, x5, x6}, {x1, x3, x2, x5, x4, x6},
      {x1, x3, x2, x6, x4, x5}, {x1, x4, x2, x3, x5, x6},
      {x1, x4, x2, x5, x3, x6}, {x1, x4, x2, x6, x3, x5},
      {x1, x5, x2, x3, x4, x6}, {x1, x5, x2, x4, x3, x6},
      {x1, x5, x2, x6, x3, x4}, {x1, x6, x2, x3, x4, x5},
      {x1, x6, x2, x4, x3, x5}, {x1, x6, x2, x5, x3, x4}};
```

次に「 $f(x) = x^6 - a_1 x^5 + a_2 x^4 - a_3 x^3 + a_4 x^2 - a_5 x + a_6$ 」と置きます.

即ち「 $f(x)$ の n 次対称式が a_n ($n = 1, 2, \dots, 6$)」です.

そして「 $\theta = \theta_a = x_1 x_2 + x_3 x_4 + x_5 x_6$ 」と取ります. このとき

```
In[*]:= p = Product[(x - (x1 x2 + x3 x4 + x5 x6)) /. AssociationThread[Σ15[[1]] → Σ15[[k]]], {k, 1, 15}]
Out[*]= (x - x3 x4 - x2 x5 - x1 x6) (x - x2 x4 - x3 x5 - x1 x6) (x - x2 x3 - x4 x5 - x1 x6)
(x - x3 x4 - x1 x5 - x2 x6) (x - x1 x4 - x3 x5 - x2 x6) (x - x1 x3 - x4 x5 - x2 x6)
(x - x2 x4 - x1 x5 - x3 x6) (x - x1 x4 - x2 x5 - x3 x6) (x - x1 x2 - x4 x5 - x3 x6)
(x - x2 x3 - x1 x5 - x4 x6) (x - x1 x3 - x2 x5 - x4 x6) (x - x1 x2 - x3 x5 - x4 x6)
(x - x2 x3 - x1 x4 - x5 x6) (x - x1 x3 - x2 x4 - x5 x6) (x - x1 x2 - x3 x4 - x5 x6)
```

これを展開すると $x_1 \sim x_6$ の対称式となるので,

SymmetricReduction を使って, $a_1 \sim a_6$ の式に直して p15a と名付けます.

`SymmetricReduction[f, {x1, ..., xn}, {s1, ..., sn}];`

`p` における初等対称式を `s1, ..., sn` で置き換えたペア `{p, q}` を与える.

但し計算には3分ほど掛かるので、実際のコマンドはコメントアウトし、結果のみを下の closed cell に入れて有ります. Mathematicaでは右下の小さいコ印をクリックして「評価」して下さい. また結果の式を見るには「セル」→「セルのプロパティ」→「開く」を選択してください. Mathematicaをお持ちでない方は「sexticFormulas.txt」をご覧ください. 結果は信じられないほど長く検索が必要な位です.

```
In[*]:= makeP15[p_] := SymmetricReduction[
    Expand[Product[(x - p) /. AssociationThread[Σ15[[1]] → Σ15[[k]], {k, 1, 15}]],
    {x1, x2, x3, x4, x5, x6}, {a1, a2, a3, a4, a5, a6}][[1]] // Collect[#, x] &;
```

```
In[*]:= (*p15a=makeP15[x1 x2+x3 x4+x5 x6]*) (*結果は、直下のclosed cellの中*)
```

p15aは

$$\theta_a = x_1 x_2 + x_3 x_4 + x_5 x_6$$

でした. p15aに具体的な数字を代入するのは後回しにして, G_{48} 不変な式を更に考えます.

$$\theta_b = x_1 x_2 x_3 x_4 + x_3 x_4 x_5 x_6 + x_5 x_6 x_1 x_2$$

$$\theta_c = x_1 x_2 (x_1 + x_2) + x_3 x_4 (x_3 + x_4) + x_5 x_6 (x_5 + x_6)$$

$$\theta_d = (x_1 + x_2)(x_3 + x_4)(x_5 + x_6)$$

$$\theta_e = (x_1 + x_2)(x_3 + x_4) + (x_3 + x_4)(x_5 + x_6) + (x_5 + x_6)(x_1 + x_2)$$

これらについても「 θ を $\Sigma 15$ の要素で置換した式の積」と「基本対称式への変形」を行い, 順に多項式 p15b, p15c, p15d, p15e を作りますが, これらの計算は p15a より更に時間がかかるので (p15cは計算に3日掛かりました), コマンドはコメントアウトし, 結果は closed cell に入れてあります. Mathematicaの場合は全て「評価」して下さい.

```
In[*]:= (*p15b=makeP15[x1 x2 x3 x4+x3 x4 x5 x6+x5 x6 x1 x2]*) (*結果は、直下のclosed cellの中*)
```

```
In[*]:= (*p15c=makeP15[x1 x2 (x1+x2)+x3 x4 (x3+ x4)+x5 x6 (x5+x6)]*)
```

```
In[*]:= (*p15d=makeP15[(x1+x2)(x3+x4)(x5+x6)]*)
```

```
In[*]:= (*p15e=makeP15[(x1+x2)(x3+x4)+(x3+x4)(x5+x6)+(x5+x6)(x1+x2)]*)
```

今までは一般の話でしたが,

ここから具体的に「例: $f(x) = x^6 + x^4 + 2x^3 + 3x^2 + 6x + 4$ 」に対して,

p15a, p15b, p15c, p15d, p15e を作り, 名前を f15a,

f15b, f15c mf15d, f15e と変え, その有理数解を求めます.

$$f = x^6 + x^4 + 2x^3 + 3x^2 + 6x + 4;$$

$$\text{alist} = \text{CoefficientList}[f, x][[1 ;; 6]] \{1, -1, 1, -1, 1, -1\} // \text{Reverse};$$

```

f15a = p15a /. AssociationThread[{a1, a2, a3, a4, a5, a6} → alist]
f15b = p15b /. AssociationThread[{a1, a2, a3, a4, a5, a6} → alist]
f15c = p15c /. AssociationThread[{a1, a2, a3, a4, a5, a6} → alist]
f15d = p15d /. AssociationThread[{a1, a2, a3, a4, a5, a6} → alist]
f15e = p15e /. AssociationThread[{a1, a2, a3, a4, a5, a6} → alist]

```

Out[*]=

$$-541696 + 2113884x + 638703x^2 - 1059033x^3 + 728599x^4 - 206759x^5 + 47539x^6 + 1839x^7 - 22118x^8 + 1442x^9 + 1065x^{10} + 609x^{11} - 121x^{12} - 15x^{13} - 3x^{14} + x^{15}$$

Out[*]=

$$6029312 - 105299968x + 185647104x^2 - 73353216x^3 - 14121216x^4 + 13377536x^5 - 6406592x^6 + 1491904x^7 - 677664x^8 + 168176x^9 - 16620x^{10} + 5140x^{11} - 831x^{12} + 39x^{13} - 9x^{14} + x^{15}$$

Out[*]=

$$104857600 - 470548480x + 779403264x^2 - 627459072x^3 + 273632256x^4 - 73699968x^5 + 18275206x^6 - 5606543x^7 + 1304848x^8 - 188436x^9 + 39292x^{10} - 11498x^{11} + 1536x^{12} + 12x^{13} - 18x^{14} + x^{15}$$

Out[*]=

$$-6229504 - 14862784x + 2989632x^2 + 6049008x^3 - 9242368x^4 + 507484x^5 + 3525684x^6 - 866575x^7 - 168248x^8 + 75652x^9 + 27104x^{10} - 1466x^{11} - 1064x^{12} - 72x^{13} + 12x^{14} + x^{15}$$

Out[*]=

$$-1703936 + 2244608x - 22016x^2 - 117824x^3 + 541952x^4 - 508224x^5 + 74304x^6 + 80144x^7 - 57496x^8 + 7241x^9 + 4512x^{10} - 1740x^{11} + 134x^{12} + 48x^{13} - 12x^{14} + x^{15}$$

f15a, f15b, f15c, f15d, f15e を有理数の範囲で因数分解します.

```

In[*]:= Factor[f15a]
Factor[f15b]
Factor[f15c]
Factor[f15d]
Factor[f15e]

```

Out[*]=

$$(1+x)(92-357x-58x^2-87x^3-7x^4+x^6)(-5888+6017x-2286x^2-234x^3+401x^4-51x^5-4x^6-4x^7+x^8)$$

Out[*]=

$$(2+x)(256-3920x-856x^2-352x^3-2x^4-x^5+x^6)(11776-31232x+29504x^2-13912x^3+3920x^4-568x^5+53x^6-10x^7+x^8)$$

Out[*]=

$$(-1+x)(2048-2048x+456x^2+19x^3+3x^4-7x^5+x^6)(-51200+127360x-63248x^2+13238x^3-3027x^4+996x^5-78x^6-10x^7+x^8)$$

Out[*]=

$$(1+x)(-184-268x+322x^2-77x^3-7x^4+5x^5+x^6)(33856-2392x-436x^2+12574x^3+573x^4-332x^5-106x^6+6x^7+x^8)$$

Out[*]=

$$(-2+x)(-416+756x-346x^2+95x^3+8x^4-6x^5+x^6)(-2048-2048x-1208x^2-696x^3+16x^4+103x^5-4x^6-4x^7+x^8)$$

「f15a = 0」の有理数解は「x = -1」なので,

「x₁ x₂ + x₃ x₄ + x₅ x₆ = -1」となります. 同様にf15b, f15c, f15d,

f15e の有理数解は, 順に -2, 1, -1, 2 となるので, 対応するθの値が求まります.

§3. 解を求める

§2の結果より

$$\begin{cases} \theta a: & x_1 x_2 + x_3 x_4 + x_5 x_6 = -1 \\ \theta b: & x_1 x_2 x_3 x_4 + x_3 x_4 x_5 x_6 + x_5 x_6 x_1 x_2 = -2 \\ \theta c: & x_1 x_2 (x_1 + x_2) + x_3 x_4 (x_3 + x_4) + x_5 x_6 (x_5 + x_6) = 1 \\ \theta d: & (x_1 + x_2) (x_3 + x_4) (x_5 + x_6) = -1 \\ \theta e: & (x_1 + x_2) (x_3 + x_4) + (x_3 + x_4) (x_5 + x_6) + (x_5 + x_6) (x_1 + x_2) = 2 \end{cases}$$

また、解と係数の関係より、以下の対称式の値はすぐ求まります。

$$\begin{cases} s_6: & (x_1 x_2) (x_3 x_4) (x_5 x_6) = a_6 = 4 \\ s_1: & (x_1 + x_2) + (x_3 + x_4) + (x_5 + x_6) = a_1 = 0 \\ s_5: & x_1 x_2 x_3 x_4 (x_5 + x_6) + x_3 x_4 x_5 x_6 (x_1 + x_2) + x_5 x_6 x_1 x_2 (x_3 + x_4) = a_5 = -6 \end{cases}$$

「3 次方程式 → 2 次方程式」の順で解いていきます。

§3-1. 3 次方程式を解く [$x_1 x_2, x_3 x_4, x_5 x_6$ と $x_1 + x_2, x_3 + x_4, x_5 + x_6$ を求める]

$\theta a, \theta b, s_6$ より、 $x_1 x_2, x_3 x_4, x_5 x_6$ は $t^3 + t^2 - 2t - 4 = 0$ の3つの解となります。これらを順に α, β, γ とおくと、次の様に順序を決めて良いです。

`In[*]:= {α, β, γ} = t /. Solve[t^3 + t^2 - 2 t - 4 == 0, t, Cubics -> True]`

`Out[*]=`

$$\left\{ \frac{1}{3} \left(-1 + (44 - 3\sqrt{177})^{1/3} + (44 + 3\sqrt{177})^{1/3} \right), \right. \\ \left. -\frac{1}{3} - \frac{1}{6} (1 + \sqrt{3}) (44 - 3\sqrt{177})^{1/3} - \frac{1}{6} (1 - \sqrt{3}) (44 + 3\sqrt{177})^{1/3}, \right. \\ \left. -\frac{1}{3} - \frac{1}{6} (1 - \sqrt{3}) (44 - 3\sqrt{177})^{1/3} - \frac{1}{6} (1 + \sqrt{3}) (44 + 3\sqrt{177})^{1/3} \right\}$$

さらに $u_1 = x_1 + x_2, u_2 = x_3 + x_4, u_3 = x_5 + x_6$ とおくと、 $s_1, \theta c, s_5$ より

$$\begin{cases} s_1: & u_1 + u_2 + u_3 = 0 \\ \theta c: & \alpha u_1 + \beta u_2 + \gamma u_3 = 1 \\ s_5: & \beta \gamma u_1 + \gamma \alpha u_2 + \alpha \beta u_3 = -6 \end{cases}$$

u_1, u_2, u_3 は上の連立一次方程式の解となるので、次の様にして求まります。

```
matrix = {{1, 1, 1}, {α, β, γ}, {β γ, γ α, α β}};
{u1, u2, u3} = LinearSolve[matrix, {0, 1, -6}] // Simplify
```

```
Out[*]:=
```

$$\left\{ \left(-354 (3i + \sqrt{59}) + 19 \sqrt{59} (1 + i \sqrt{3}) (44 + 3 \sqrt{177})^{2/3} + \sqrt{59} (1 - i \sqrt{3}) (44 - 3 \sqrt{177})^{1/3} \right. \right. \\ \left. \left. (44 + 3 \sqrt{177})^{2/3} + \sqrt{59} (44 - 3 \sqrt{177})^{2/3} (19 - 19i \sqrt{3} + (1 + i \sqrt{3}) (44 + 3 \sqrt{177})^{1/3}) \right) \right. \\ \left. \left(354 ((-3i + \sqrt{3}) (44 - 3 \sqrt{177})^{1/3} - (3i + \sqrt{3}) (44 + 3 \sqrt{177})^{1/3}) \right), \right. \\ \left. - \left((177 (-3i + \sqrt{59}) + 19 \sqrt{59} (44 + 3 \sqrt{177})^{2/3} + \sqrt{59} (44 - 3 \sqrt{177})^{1/3} (44 + 3 \sqrt{177})^{2/3} + \right. \right. \\ \left. \left. \sqrt{59} (44 - 3 \sqrt{177})^{2/3} (19 + (44 + 3 \sqrt{177})^{1/3}) \right) \right. \\ \left. \left(177 ((-3i + \sqrt{3}) (44 - 3 \sqrt{177})^{1/3} - (3i + \sqrt{3}) (44 + 3 \sqrt{177})^{1/3}) \right) \right) \right. \\ \left. \frac{1}{36 \sqrt{59}} \left(-19 (-3i + \sqrt{3}) (44 - 3 \sqrt{177})^{1/3} + (3i + \sqrt{3}) (44 - 3 \sqrt{177})^{2/3} + \right. \right. \\ \left. \left. 19 (3i + \sqrt{3}) (44 + 3 \sqrt{177})^{1/3} - (-3i + \sqrt{3}) (44 + 3 \sqrt{177})^{2/3} \right) \right\}$$

§3-2. 2 次方程式を3組解く [(x₁ + x₂, x₁ x₂), (x₃ + x₄, x₃ x₄), (x₅ + x₆, x₅ x₆) の値から x₁~x₆を求める]

以上より (x₁ + x₂, x₁ x₂), (x₃ + x₄, x₃ x₄), (x₅ + x₆, x₅ x₆) 3 組の値 (u₁, α), (u₂, β), (u₃, γ) が全て求まりました。一般に x + y = u,

xy = v のとき {x, y} = $\frac{1}{2} \left\{ u + \sqrt{u^2 - 4v}, u - \sqrt{u^2 - 4v} \right\}$ なので,

x₁~x₆ の値は次の様に求まります。(もちろんSolveを使って求められるのですが, その場合, 3 重根の中に全ての項が入って見づらく感じたので, この様にしています.)

```
In[*]:= solve[u_, v_] = 1 / 2 { (u + Sqrt[u^2 - 4 v]), (u - Sqrt[u^2 - 4 v]) };
```

```
In[*]:= {x1, x2, x3, x4, x5, x6} = {solve[u1, α], solve[u2, β], solve[u3, γ]} // Flatten;
```

x₁ のみ表示すると次の様になります。

```
In[*]:= x1 // TraditionalForm
```

```
Out[*]//TraditionalForm=
```

$$\frac{1}{2} \left(\left(-354 (\sqrt{59} + 3i) + 19 \sqrt{59} (1 + i \sqrt{3}) (44 + 3 \sqrt{177})^{2/3} + \sqrt{59} (1 - i \sqrt{3}) \sqrt[3]{44 - 3 \sqrt{177}} (44 + 3 \sqrt{177})^{2/3} + \right. \right. \\ \left. \left. \sqrt{59} (44 - 3 \sqrt{177})^{2/3} (19 - 19i \sqrt{3} + (1 + i \sqrt{3}) \sqrt[3]{44 + 3 \sqrt{177}}) \right) \right. \\ \left. (354 ((\sqrt{3} - 3i) \sqrt[3]{44 - 3 \sqrt{177}} - (\sqrt{3} + 3i) \sqrt[3]{44 + 3 \sqrt{177}})) + \right. \\ \left. \sqrt{\left(-\frac{4}{3} (-1 + \sqrt[3]{44 - 3 \sqrt{177}} + \sqrt[3]{44 + 3 \sqrt{177}}) + \right. \right. \\ \left. \left. (-354 (\sqrt{59} + 3i) + 19 \sqrt{59} (1 + i \sqrt{3}) (44 + 3 \sqrt{177})^{2/3} + \sqrt{59} (1 - i \sqrt{3}) \sqrt[3]{44 - 3 \sqrt{177}} \right. \right. \\ \left. \left. (44 + 3 \sqrt{177})^{2/3} + \sqrt{59} (44 - 3 \sqrt{177})^{2/3} (19 - 19i \sqrt{3} + (1 + i \sqrt{3}) \sqrt[3]{44 + 3 \sqrt{177}}) \right)^2 \right. \\ \left. \left. (125316 ((\sqrt{3} - 3i) \sqrt[3]{44 - 3 \sqrt{177}} - (\sqrt{3} + 3i) \sqrt[3]{44 + 3 \sqrt{177}})^2) \right) \right)$$

検算です。最初は厳密解の近似値、次はNSolveを使った近似解です。ご覧のように一致します。

```
In[*]:= N[{x1, x2, x3, x4, x5, x6}] // Sort
Out[*]=
{-0.902149 + 0.256938 i, -0.902149 - 0.256938 i, -0.226699 + 1.2679 i,
-0.226699 - 1.2679 i, 1.12885 - 1.21077 i, 1.12885 + 1.21077 i}

In[*]:= x /. NSolve[f == 0, x] // Sort
Out[*]=
{-0.902149 - 0.256938 i, -0.902149 + 0.256938 i, -0.226699 - 1.2679 i,
-0.226699 + 1.2679 i, 1.12885 - 1.21077 i, 1.12885 + 1.21077 i}
```

§4. 解を求める(続き)

A. §3の例では α, β, γ がすべて異なっていたので、下の3式が一次独立となり、 u_1, u_2, u_3 の値が求まりました。(因みに下の行列の行列式は $\text{Det} = (\alpha - \beta)(\beta - \gamma)(\gamma - \alpha)$ です。)

$$\begin{cases} a_1 = & u_1 + u_2 + u_3 \\ \theta c = & \alpha u_1 + \beta u_2 + \gamma u_3 \\ a_5 = & \beta \gamma u_1 + \gamma \alpha u_2 + \alpha \beta u_3 \end{cases}$$

B. $\alpha = \beta \neq \gamma$ のとき、上の式から $u_3 = \theta c / (\gamma - \alpha)$ 、また「 $\theta e = u_1 u_2 + u_2 u_3 + u_3 u_1$ 」と組み合わせると、 u_1, u_2 は「 $u^2 - (a_1 - u_3)u + (\theta e + u_3^2) = 0$ 」の2解となります。

C. $\alpha = \beta = \gamma$ のとき、 $a_1 = u_1 + u_2 + u_3$ 、 $\theta e = u_1 u_2 + u_2 u_3 + u_3 u_1$ 、 $d = u_1 u_2 u_3$ より、 u_1, u_2, u_3 は「 $u^3 - a_1 u + (\theta e)u - (\theta d) = 0$ 」の3解となります。

(これ以外のケースでこの3方程式を使わなかったのは α, β, γ と対応する u_1, u_2, u_3 を選ぶのが面倒だったせいですが、 $\alpha = \beta = \gamma$ のときはどの組み合わせも本質的に同じとなります。)

実は $\text{gal} = G_{48}$ の方程式については f15a~f15e の有理数解が複数個存在することが多々あり、これは大問題ですが、それについては solveSexticProgram.nb をご覧下さい。